// File: @chainlink/contracts/src/v0.8/interfaces/AggregatorV3Interface.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;

interface AggregatorV3Interface {
  function decimals() external view returns (uint8);

  function description() external view returns (string memory);

  function version() external view returns (uint256);

  function getRoundData(uint80 _roundId)
    external
    view
    returns (
      uint80 roundId,
      int256 answer,
      uint256 startedAt,
      uint256 updatedAt,
      uint80 answeredInRound
    );

  function latestRoundData()
    external
    view
    returns (
      uint80 roundId,
      int256 answer,
      uint256 startedAt,
      uint256 updatedAt,
      uint80 answeredInRound
    );
}

// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/security/Pausable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (security/Pausable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which allows children to implement an emergency stop
 * mechanism that can be triggered by an authorized account.
 *
 * This module is used through inheritance. It will make available the
 * modifiers `whenNotPaused` and `whenPaused`, which can be applied to
 * the functions of your contract. Note that they will not be pausable by
 * simply including this module, only once the modifiers are put in place.
 */
abstract contract Pausable is Context {
    /**
     * @dev Emitted when the pause is triggered by `account`.
     */
    event Paused(address account);

    /**
     * @dev Emitted when the pause is lifted by `account`.
     */
    event Unpaused(address account);

    bool private _paused;

    /**
     * @dev Initializes the contract in unpaused state.
     */
    constructor() {
        _paused = false;
    }

    /**
     * @dev Modifier to make a function callable only when the contract is not paused.
     *
     * Requirements:
     *
     * - The contract must not be paused.
     */
    modifier whenNotPaused() {
        _requireNotPaused();
        _;
    }

    /**
     * @dev Modifier to make a function callable only when the contract is paused.
     *
     * Requirements:
     *
     * - The contract must be paused.
     */
    modifier whenPaused() {
        _requirePaused();
        _;
    }

    /**
     * @dev Returns true if the contract is paused, and false otherwise.
     */
    function paused() public view virtual returns (bool) {
        return _paused;
    }

    /**
     * @dev Throws if the contract is paused.
     */
    function _requireNotPaused() internal view virtual {
        require(!paused(), "Pausable: paused");
    }

    /**
     * @dev Throws if the contract is not paused.
     */
    function _requirePaused() internal view virtual {
        require(paused(), "Pausable: not paused");
    }

    /**
     * @dev Triggers stopped state.
     *
     * Requirements:
     *
     * - The contract must not be paused.
     */
    function _pause() internal virtual whenNotPaused {
        _paused = true;
        emit Paused(_msgSender());
    }

    /**
     * @dev Returns to normal state.
     *
     * Requirements:
     *
     * - The contract must be paused.
     */
    function _unpause() internal virtual whenPaused {
        _paused = false;
        emit Unpaused(_msgSender());
    }
}

// File: @openzeppelin/contracts/token/ERC1155/ERC1155.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (token/ERC1155/ERC1155.sol)

pragma solidity ^0.8.0;

import "./IERC1155.sol";
import "./IERC1155Receiver.sol";
import "./extensions/IERC1155MetadataURI.sol";
import "../../utils/Address.sol";
import "../../utils/Context.sol";
import "../../utils/introspection/ERC165.sol";

/**
 * @dev Implementation of the basic standard multi-token.
 * See https://eips.ethereum.org/EIPS/eip-1155
 * Originally based on code by Enjin: https://github.com/enjin/erc-1155
 *
 * _Available since v3.1._
 */
contract ERC1155 is Context, ERC165, IERC1155, IERC1155MetadataURI {
    using Address for address;

    // Mapping from token ID to account balances
    mapping(uint256 => mapping(address => uint256)) private _balances;

    // Mapping from account to operator approvals
    mapping(address => mapping(address => bool)) private _operatorApprovals;

    // Used as the URI for all token types by relying on ID substitution, e.g. https://token-cdn-domain/{id}.json
    string private _uri;

    /**
     * @dev See {_setURI}.
     */
    constructor(string memory uri_) {
        _setURI(uri_);
    }

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override(ERC165, IERC165) returns (bool) {
        return
            interfaceId == type(IERC1155).interfaceId ||
            interfaceId == type(IERC1155MetadataURI).interfaceId ||
            super.supportsInterface(interfaceId);
    }

    /**
     * @dev See {IERC1155MetadataURI-uri}.
     *
     * This implementation returns the same URI for *all* token types. It relies
     * on the token type ID substitution mechanism
     * https://eips.ethereum.org/EIPS/eip-1155#metadata[defined in the EIP].
     *
     * Clients calling this function must replace the `\{id\}` substring with the
     * actual token type ID.
     */
    function uri(uint256) public view virtual override returns (string memory) {
        return _uri;
    }

    /**
     * @dev See {IERC1155-balanceOf}.
     *
     * Requirements:
     *
     * - `account` cannot be the zero address.
     */
    function balanceOf(address account, uint256 id) public view virtual override returns (uint256) {
        require(account != address(0), "ERC1155: address zero is not a valid owner");
        return _balances[id][account];
    }

    /**
     * @dev See {IERC1155-balanceOfBatch}.
     *
     * Requirements:
     *
     * - `accounts` and `ids` must have the same length.
     */
    function balanceOfBatch(address[] memory accounts, uint256[] memory ids)
        public
        view
        virtual
        override
        returns (uint256[] memory)
    {
        require(accounts.length == ids.length, "ERC1155: accounts and ids length mismatch");

        uint256[] memory batchBalances = new uint256[](accounts.length);

        for (uint256 i = 0; i < accounts.length; ++i) {
            batchBalances[i] = balanceOf(accounts[i], ids[i]);
        }

        return batchBalances;
    }

    /**
     * @dev See {IERC1155-setApprovalForAll}.
     */
    function setApprovalForAll(address operator, bool approved) public virtual override {
        _setApprovalForAll(_msgSender(), operator, approved);
    }

    /**
     * @dev See {IERC1155-isApprovedForAll}.
     */
    function isApprovedForAll(address account, address operator) public view virtual override returns (bool) {
        return _operatorApprovals[account][operator];
    }

    /**
     * @dev See {IERC1155-safeTransferFrom}.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 id,
        uint256 amount,
        bytes memory data
    ) public virtual override {
        require(
            from == _msgSender() || isApprovedForAll(from, _msgSender()),
            "ERC1155: caller is not token owner or approved"
        );
        _safeTransferFrom(from, to, id, amount, data);
    }

    /**
     * @dev See {IERC1155-safeBatchTransferFrom}.
     */
    function safeBatchTransferFrom(
        address from,
        address to,
        uint256[] memory ids,
        uint256[] memory amounts,
        bytes memory data
    ) public virtual override {
        require(
            from == _msgSender() || isApprovedForAll(from, _msgSender()),
            "ERC1155: caller is not token owner or approved"
        );
        _safeBatchTransferFrom(from, to, ids, amounts, data);
    }

    /**
     * @dev Transfers `amount` tokens of token type `id` from `from` to `to`.
     *
     * Emits a {TransferSingle} event.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     * - `from` must have a balance of tokens of type `id` of at least `amount`.
     * - If `to` refers to a smart contract, it must implement {IERC1155Receiver-onERC1155Received} and return the
     * acceptance magic value.
     */
    function _safeTransferFrom(
        address from,
        address to,
        uint256 id,
        uint256 amount,
        bytes memory data
    ) internal virtual {
        require(to != address(0), "ERC1155: transfer to the zero address");

        address operator = _msgSender();
        uint256[] memory ids = _asSingletonArray(id);
        uint256[] memory amounts = _asSingletonArray(amount);

        _beforeTokenTransfer(operator, from, to, ids, amounts, data);

        uint256 fromBalance = _balances[id][from];
        require(fromBalance >= amount, "ERC1155: insufficient balance for transfer");
        unchecked {
            _balances[id][from] = fromBalance - amount;
        }
        _balances[id][to] += amount;

        emit TransferSingle(operator, from, to, id, amount);

        _afterTokenTransfer(operator, from, to, ids, amounts, data);

        _doSafeTransferAcceptanceCheck(operator, from, to, id, amount, data);
    }

    /**
     * @dev xref:ROOT:erc1155.adoc#batch-operations[Batched] version of {_safeTransferFrom}.
     *
     * Emits a {TransferBatch} event.
     *
     * Requirements:
     *
     * - If `to` refers to a smart contract, it must implement {IERC1155Receiver-onERC1155BatchReceived} and return the
     * acceptance magic value.
     */
    function _safeBatchTransferFrom(
        address from,
        address to,
        uint256[] memory ids,
        uint256[] memory amounts,
        bytes memory data
    ) internal virtual {
        require(ids.length == amounts.length, "ERC1155: ids and amounts length mismatch");
        require(to != address(0), "ERC1155: transfer to the zero address");

        address operator = _msgSender();

        _beforeTokenTransfer(operator, from, to, ids, amounts, data);

        for (uint256 i = 0; i < ids.length; ++i) {
            uint256 id = ids[i];
            uint256 amount = amounts[i];

            uint256 fromBalance = _balances[id][from];
            require(fromBalance >= amount, "ERC1155: insufficient balance for transfer");
            unchecked {
                _balances[id][from] = fromBalance - amount;
            }
            _balances[id][to] += amount;
        }

        emit TransferBatch(operator, from, to, ids, amounts);

        _afterTokenTransfer(operator, from, to, ids, amounts, data);

        _doSafeBatchTransferAcceptanceCheck(operator, from, to, ids, amounts, data);
    }

    /**
     * @dev Sets a new URI for all token types, by relying on the token type ID
     * substitution mechanism
     * https://eips.ethereum.org/EIPS/eip-1155#metadata[defined in the EIP].
     *
     * By this mechanism, any occurrence of the `\{id\}` substring in either the
     * URI or any of the amounts in the JSON file at said URI will be replaced by
     * clients with the token type ID.
     *
     * For example, the `https://token-cdn-domain/\{id\}.json` URI would be
     * interpreted by clients as
     * `https://token-cdn-domain/000000000000000000000000000000000000000000000000000000000004cce0.json`
     * for token type ID 0x4cce0.
     *
     * See {uri}.
     *
     * Because these URIs cannot be meaningfully represented by the {URI} event,
     * this function emits no events.
     */
    function _setURI(string memory newuri) internal virtual {
        _uri = newuri;
    }

    /**
     * @dev Creates `amount` tokens of token type `id`, and assigns them to `to`.
     *
     * Emits a {TransferSingle} event.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     * - If `to` refers to a smart contract, it must implement {IERC1155Receiver-onERC1155Received} and return the
     * acceptance magic value.
     */
    function _mint(
        address to,
        uint256 id,
        uint256 amount,
        bytes memory data
    ) internal virtual {
        require(to != address(0), "ERC1155: mint to the zero address");

        address operator = _msgSender();
        uint256[] memory ids = _asSingletonArray(id);
        uint256[] memory amounts = _asSingletonArray(amount);

        _beforeTokenTransfer(operator, address(0), to, ids, amounts, data);

        _balances[id][to] += amount;
        emit TransferSingle(operator, address(0), to, id, amount);

        _afterTokenTransfer(operator, address(0), to, ids, amounts, data);

        _doSafeTransferAcceptanceCheck(operator, address(0), to, id, amount, data);
    }

    /**
     * @dev xref:ROOT:erc1155.adoc#batch-operations[Batched] version of {_mint}.
     *
     * Emits a {TransferBatch} event.
     *
     * Requirements:
     *
     * - `ids` and `amounts` must have the same length.
     * - If `to` refers to a smart contract, it must implement {IERC1155Receiver-onERC1155BatchReceived} and return the
     * acceptance magic value.
     */
    function _mintBatch(
        address to,
        uint256[] memory ids,
        uint256[] memory amounts,
        bytes memory data
    ) internal virtual {
        require(to != address(0), "ERC1155: mint to the zero address");
        require(ids.length == amounts.length, "ERC1155: ids and amounts length mismatch");

        address operator = _msgSender();

        _beforeTokenTransfer(operator, address(0), to, ids, amounts, data);

        for (uint256 i = 0; i < ids.length; i++) {
            _balances[ids[i]][to] += amounts[i];
        }

        emit TransferBatch(operator, address(0), to, ids, amounts);

        _afterTokenTransfer(operator, address(0), to, ids, amounts, data);

        _doSafeBatchTransferAcceptanceCheck(operator, address(0), to, ids, amounts, data);
    }

    /**
     * @dev Destroys `amount` tokens of token type `id` from `from`
     *
     * Emits a {TransferSingle} event.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `from` must have at least `amount` tokens of token type `id`.
     */
    function _burn(
        address from,
        uint256 id,
        uint256 amount
    ) internal virtual {
        require(from != address(0), "ERC1155: burn from the zero address");

        address operator = _msgSender();
        uint256[] memory ids = _asSingletonArray(id);
        uint256[] memory amounts = _asSingletonArray(amount);

        _beforeTokenTransfer(operator, from, address(0), ids, amounts, "");

        uint256 fromBalance = _balances[id][from];
        require(fromBalance >= amount, "ERC1155: burn amount exceeds balance");
        unchecked {
            _balances[id][from] = fromBalance - amount;
        }

        emit TransferSingle(operator, from, address(0), id, amount);

        _afterTokenTransfer(operator, from, address(0), ids, amounts, "");
    }

    /**
     * @dev xref:ROOT:erc1155.adoc#batch-operations[Batched] version of {_burn}.
     *
     * Emits a {TransferBatch} event.
     *
     * Requirements:
     *
     * - `ids` and `amounts` must have the same length.
     */
    function _burnBatch(
        address from,
        uint256[] memory ids,
        uint256[] memory amounts
    ) internal virtual {
        require(from != address(0), "ERC1155: burn from the zero address");
        require(ids.length == amounts.length, "ERC1155: ids and amounts length mismatch");

        address operator = _msgSender();

        _beforeTokenTransfer(operator, from, address(0), ids, amounts, "");

        for (uint256 i = 0; i < ids.length; i++) {
            uint256 id = ids[i];
            uint256 amount = amounts[i];

            uint256 fromBalance = _balances[id][from];
            require(fromBalance >= amount, "ERC1155: burn amount exceeds balance");
            unchecked {
                _balances[id][from] = fromBalance - amount;
            }
        }

        emit TransferBatch(operator, from, address(0), ids, amounts);

        _afterTokenTransfer(operator, from, address(0), ids, amounts, "");
    }

    /**
     * @dev Approve `operator` to operate on all of `owner` tokens
     *
     * Emits an {ApprovalForAll} event.
     */
    function _setApprovalForAll(
        address owner,
        address operator,
        bool approved
    ) internal virtual {
        require(owner != operator, "ERC1155: setting approval status for self");
        _operatorApprovals[owner][operator] = approved;
        emit ApprovalForAll(owner, operator, approved);
    }

    /**
     * @dev Hook that is called before any token transfer. This includes minting
     * and burning, as well as batched variants.
     *
     * The same hook is called on both single and batched variants. For single
     * transfers, the length of the `ids` and `amounts` arrays will be 1.
     *
     * Calling conditions (for each `id` and `amount` pair):
     *
     * - When `from` and `to` are both non-zero, `amount` of ``from``'s tokens
     * of token type `id` will be  transferred to `to`.
     * - When `from` is zero, `amount` tokens of token type `id` will be minted
     * for `to`.
     * - when `to` is zero, `amount` of ``from``'s tokens of token type `id`
     * will be burned.
     * - `from` and `to` are never both zero.
     * - `ids` and `amounts` have the same, non-zero length.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(
        address operator,
        address from,
        address to,
        uint256[] memory ids,
        uint256[] memory amounts,
        bytes memory data
    ) internal virtual {}

    /**
     * @dev Hook that is called after any token transfer. This includes minting
     * and burning, as well as batched variants.
     *
     * The same hook is called on both single and batched variants. For single
     * transfers, the length of the `id` and `amount` arrays will be 1.
     *
     * Calling conditions (for each `id` and `amount` pair):
     *
     * - When `from` and `to` are both non-zero, `amount` of ``from``'s tokens
     * of token type `id` will be  transferred to `to`.
     * - When `from` is zero, `amount` tokens of token type `id` will be minted
     * for `to`.
     * - when `to` is zero, `amount` of ``from``'s tokens of token type `id`
     * will be burned.
     * - `from` and `to` are never both zero.
     * - `ids` and `amounts` have the same, non-zero length.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _afterTokenTransfer(
        address operator,
        address from,
        address to,
        uint256[] memory ids,
        uint256[] memory amounts,
        bytes memory data
    ) internal virtual {}

    function _doSafeTransferAcceptanceCheck(
        address operator,
        address from,
        address to,
        uint256 id,
        uint256 amount,
        bytes memory data
    ) private {
        if (to.isContract()) {
            try IERC1155Receiver(to).onERC1155Received(operator, from, id, amount, data) returns (bytes4 response) {
                if (response != IERC1155Receiver.onERC1155Received.selector) {
                    revert("ERC1155: ERC1155Receiver rejected tokens");
                }
            } catch Error(string memory reason) {
                revert(reason);
            } catch {
                revert("ERC1155: transfer to non-ERC1155Receiver implementer");
            }
        }
    }

    function _doSafeBatchTransferAcceptanceCheck(
        address operator,
        address from,
        address to,
        uint256[] memory ids,
        uint256[] memory amounts,
        bytes memory data
    ) private {
        if (to.isContract()) {
            try IERC1155Receiver(to).onERC1155BatchReceived(operator, from, ids, amounts, data) returns (
                bytes4 response
            ) {
                if (response != IERC1155Receiver.onERC1155BatchReceived.selector) {
                    revert("ERC1155: ERC1155Receiver rejected tokens");
                }
            } catch Error(string memory reason) {
                revert(reason);
            } catch {
                revert("ERC1155: transfer to non-ERC1155Receiver implementer");
            }
        }
    }

    function _asSingletonArray(uint256 element) private pure returns (uint256[] memory) {
        uint256[] memory array = new uint256[](1);
        array[0] = element;

        return array;
    }
}

// File: @openzeppelin/contracts/token/ERC1155/extensions/ERC1155Burnable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (token/ERC1155/extensions/ERC1155Burnable.sol)

pragma solidity ^0.8.0;

import "../ERC1155.sol";

/**
 * @dev Extension of {ERC1155} that allows token holders to destroy both their
 * own tokens and those that they have been approved to use.
 *
 * _Available since v3.1._
 */
abstract contract ERC1155Burnable is ERC1155 {
    function burn(
        address account,
        uint256 id,
        uint256 value
    ) public virtual {
        require(
            account == _msgSender() || isApprovedForAll(account, _msgSender()),
            "ERC1155: caller is not token owner or approved"
        );

        _burn(account, id, value);
    }

    function burnBatch(
        address account,
        uint256[] memory ids,
        uint256[] memory values
    ) public virtual {
        require(
            account == _msgSender() || isApprovedForAll(account, _msgSender()),
            "ERC1155: caller is not token owner or approved"
        );

        _burnBatch(account, ids, values);
    }
}

// File: @openzeppelin/contracts/token/ERC1155/extensions/ERC1155Supply.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC1155/extensions/ERC1155Supply.sol)

pragma solidity ^0.8.0;

import "../ERC1155.sol";

/**
 * @dev Extension of ERC1155 that adds tracking of total supply per id.
 *
 * Useful for scenarios where Fungible and Non-fungible tokens have to be
 * clearly identified. Note: While a totalSupply of 1 might mean the
 * corresponding is an NFT, there is no guarantees that no other token with the
 * same id are not going to be minted.
 */
abstract contract ERC1155Supply is ERC1155 {
    mapping(uint256 => uint256) private _totalSupply;

    /**
     * @dev Total amount of tokens in with a given id.
     */
    function totalSupply(uint256 id) public view virtual returns (uint256) {
        return _totalSupply[id];
    }

    /**
     * @dev Indicates whether any token exist with a given id, or not.
     */
    function exists(uint256 id) public view virtual returns (bool) {
        return ERC1155Supply.totalSupply(id) > 0;
    }

    /**
     * @dev See {ERC1155-_beforeTokenTransfer}.
     */
    function _beforeTokenTransfer(
        address operator,
        address from,
        address to,
        uint256[] memory ids,
        uint256[] memory amounts,
        bytes memory data
    ) internal virtual override {
        super._beforeTokenTransfer(operator, from, to, ids, amounts, data);

        if (from == address(0)) {
            for (uint256 i = 0; i < ids.length; ++i) {
                _totalSupply[ids[i]] += amounts[i];
            }
        }

        if (to == address(0)) {
            for (uint256 i = 0; i < ids.length; ++i) {
                uint256 id = ids[i];
                uint256 amount = amounts[i];
                uint256 supply = _totalSupply[id];
                require(supply >= amount, "ERC1155: burn amount exceeds totalSupply");
                unchecked {
                    _totalSupply[id] = supply - amount;
                }
            }
        }
    }
}

// File: @openzeppelin/contracts/token/ERC1155/extensions/IERC1155MetadataURI.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC1155/extensions/IERC1155MetadataURI.sol)

pragma solidity ^0.8.0;

import "../IERC1155.sol";

/**
 * @dev Interface of the optional ERC1155MetadataExtension interface, as defined
 * in the https://eips.ethereum.org/EIPS/eip-1155#metadata-extensions[EIP].
 *
 * _Available since v3.1._
 */
interface IERC1155MetadataURI is IERC1155 {
    /**
     * @dev Returns the URI for token type `id`.
     *
     * If the `\{id\}` substring is present in the URI, it must be replaced by
     * clients with the actual token type ID.
     */
    function uri(uint256 id) external view returns (string memory);
}

// File: @openzeppelin/contracts/token/ERC1155/IERC1155.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (token/ERC1155/IERC1155.sol)

pragma solidity ^0.8.0;

import "../../utils/introspection/IERC165.sol";

/**
 * @dev Required interface of an ERC1155 compliant contract, as defined in the
 * https://eips.ethereum.org/EIPS/eip-1155[EIP].
 *
 * _Available since v3.1._
 */
interface IERC1155 is IERC165 {
    /**
     * @dev Emitted when `value` tokens of token type `id` are transferred from `from` to `to` by `operator`.
     */
    event TransferSingle(address indexed operator, address indexed from, address indexed to, uint256 id, uint256 value);

    /**
     * @dev Equivalent to multiple {TransferSingle} events, where `operator`, `from` and `to` are the same for all
     * transfers.
     */
    event TransferBatch(
        address indexed operator,
        address indexed from,
        address indexed to,
        uint256[] ids,
        uint256[] values
    );

    /**
     * @dev Emitted when `account` grants or revokes permission to `operator` to transfer their tokens, according to
     * `approved`.
     */
    event ApprovalForAll(address indexed account, address indexed operator, bool approved);

    /**
     * @dev Emitted when the URI for token type `id` changes to `value`, if it is a non-programmatic URI.
     *
     * If an {URI} event was emitted for `id`, the standard
     * https://eips.ethereum.org/EIPS/eip-1155#metadata-extensions[guarantees] that `value` will equal the value
     * returned by {IERC1155MetadataURI-uri}.
     */
    event URI(string value, uint256 indexed id);

    /**
     * @dev Returns the amount of tokens of token type `id` owned by `account`.
     *
     * Requirements:
     *
     * - `account` cannot be the zero address.
     */
    function balanceOf(address account, uint256 id) external view returns (uint256);

    /**
     * @dev xref:ROOT:erc1155.adoc#batch-operations[Batched] version of {balanceOf}.
     *
     * Requirements:
     *
     * - `accounts` and `ids` must have the same length.
     */
    function balanceOfBatch(address[] calldata accounts, uint256[] calldata ids)
        external
        view
        returns (uint256[] memory);

    /**
     * @dev Grants or revokes permission to `operator` to transfer the caller's tokens, according to `approved`,
     *
     * Emits an {ApprovalForAll} event.
     *
     * Requirements:
     *
     * - `operator` cannot be the caller.
     */
    function setApprovalForAll(address operator, bool approved) external;

    /**
     * @dev Returns true if `operator` is approved to transfer ``account``'s tokens.
     *
     * See {setApprovalForAll}.
     */
    function isApprovedForAll(address account, address operator) external view returns (bool);

    /**
     * @dev Transfers `amount` tokens of token type `id` from `from` to `to`.
     *
     * Emits a {TransferSingle} event.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     * - If the caller is not `from`, it must have been approved to spend ``from``'s tokens via {setApprovalForAll}.
     * - `from` must have a balance of tokens of type `id` of at least `amount`.
     * - If `to` refers to a smart contract, it must implement {IERC1155Receiver-onERC1155Received} and return the
     * acceptance magic value.
     */
    function safeTransferFrom(
        address from,
        address to,
        uint256 id,
        uint256 amount,
        bytes calldata data
    ) external;

    /**
     * @dev xref:ROOT:erc1155.adoc#batch-operations[Batched] version of {safeTransferFrom}.
     *
     * Emits a {TransferBatch} event.
     *
     * Requirements:
     *
     * - `ids` and `amounts` must have the same length.
     * - If `to` refers to a smart contract, it must implement {IERC1155Receiver-onERC1155BatchReceived} and return the
     * acceptance magic value.
     */
    function safeBatchTransferFrom(
        address from,
        address to,
        uint256[] calldata ids,
        uint256[] calldata amounts,
        bytes calldata data
    ) external;
}

// File: @openzeppelin/contracts/token/ERC1155/IERC1155Receiver.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.5.0) (token/ERC1155/IERC1155Receiver.sol)

pragma solidity ^0.8.0;

import "../../utils/introspection/IERC165.sol";

/**
 * @dev _Available since v3.1._
 */
interface IERC1155Receiver is IERC165 {
    /**
     * @dev Handles the receipt of a single ERC1155 token type. This function is
     * called at the end of a `safeTransferFrom` after the balance has been updated.
     *
     * NOTE: To accept the transfer, this must return
     * `bytes4(keccak256("onERC1155Received(address,address,uint256,uint256,bytes)"))`
     * (i.e. 0xf23a6e61, or its own function selector).
     *
     * @param operator The address which initiated the transfer (i.e. msg.sender)
     * @param from The address which previously owned the token
     * @param id The ID of the token being transferred
     * @param value The amount of tokens being transferred
     * @param data Additional data with no specified format
     * @return `bytes4(keccak256("onERC1155Received(address,address,uint256,uint256,bytes)"))` if transfer is allowed
     */
    function onERC1155Received(
        address operator,
        address from,
        uint256 id,
        uint256 value,
        bytes calldata data
    ) external returns (bytes4);

    /**
     * @dev Handles the receipt of a multiple ERC1155 token types. This function
     * is called at the end of a `safeBatchTransferFrom` after the balances have
     * been updated.
     *
     * NOTE: To accept the transfer(s), this must return
     * `bytes4(keccak256("onERC1155BatchReceived(address,address,uint256[],uint256[],bytes)"))`
     * (i.e. 0xbc197c81, or its own function selector).
     *
     * @param operator The address which initiated the batch transfer (i.e. msg.sender)
     * @param from The address which previously owned the token
     * @param ids An array containing ids of each token being transferred (order and length must match values array)
     * @param values An array containing amounts of each token being transferred (order and length must match ids array)
     * @param data Additional data with no specified format
     * @return `bytes4(keccak256("onERC1155BatchReceived(address,address,uint256[],uint256[],bytes)"))` if transfer is allowed
     */
    function onERC1155BatchReceived(
        address operator,
        address from,
        uint256[] calldata ids,
        uint256[] calldata values,
        bytes calldata data
    ) external returns (bytes4);
}

// File: @openzeppelin/contracts/utils/Address.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.8.0) (utils/Address.sol)

pragma solidity ^0.8.1;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     *
     * [IMPORTANT]
     * ====
     * You shouldn't rely on `isContract` to protect against flash loan attacks!
     *
     * Preventing calls from contracts is highly discouraged. It breaks composability, breaks support for smart wallets
     * like Gnosis Safe, and does not provide security since it can be circumvented by calling from a contract
     * constructor.
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize/address.code.length, which returns 0
        // for contracts in construction, since the code is only stored at the end
        // of the constructor execution.

        return account.code.length > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verify that a low level call to smart-contract was successful, and revert (either by bubbling
     * the revert reason or using the provided one) in case of unsuccessful call or if target was not a contract.
     *
     * _Available since v4.8._
     */
    function verifyCallResultFromTarget(
        address target,
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        if (success) {
            if (returndata.length == 0) {
                // only check isContract if the call was successful and the return data is empty
                // otherwise we already know that it was a contract
                require(isContract(target), "Address: call to non-contract");
            }
            return returndata;
        } else {
            _revert(returndata, errorMessage);
        }
    }

    /**
     * @dev Tool to verify that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason or using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            _revert(returndata, errorMessage);
        }
    }

    function _revert(bytes memory returndata, string memory errorMessage) private pure {
        // Look for revert reason and bubble it up if present
        if (returndata.length > 0) {
            // The easiest way to bubble the revert reason is using memory via assembly
            /// @solidity memory-safe-assembly
            assembly {
                let returndata_size := mload(returndata)
                revert(add(32, returndata), returndata_size)
            }
        } else {
            revert(errorMessage);
        }
    }
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @openzeppelin/contracts/utils/introspection/ERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/ERC165.sol)

pragma solidity ^0.8.0;

import "./IERC165.sol";

/**
 * @dev Implementation of the {IERC165} interface.
 *
 * Contracts that want to implement ERC165 should inherit from this contract and override {supportsInterface} to check
 * for the additional interface id that will be supported. For example:
 *
 * ```solidity
 * function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
 *     return interfaceId == type(MyInterface).interfaceId || super.supportsInterface(interfaceId);
 * }
 * ```
 *
 * Alternatively, {ERC165Storage} provides an easier to use but more expensive implementation.
 */
abstract contract ERC165 is IERC165 {
    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IERC165).interfaceId;
    }
}

// File: @openzeppelin/contracts/utils/introspection/IERC165.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/introspection/IERC165.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}

// File: contracts/WhoAnonByMetadrop.sol



// SPDX-License-Identifier: BUSL-1.1
pragma solidity 0.8.17;

//                      %@@@@@                @@@@@
//                      %@@@@@                @@@@@
//              @       %@@@@@       @        @@@@@       ,
//             @@@@@@   %@@@@@   @@@@@@@@@    @@@@@   .@@@@@&
//             /@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@@
//                #@@@@@@@@@@@@@@@@@    @@@@@@@@@@@@@@@@@
//                   @@@@@@@@@@@@          @@@@@@@@@@@
//              @@@@@@@@@@@@@@@@@@@@  @@@@@@@@@@@@@@@@@@@
//            @@@@@@@@@@%@@@@@@@@@@@@@@@@@@@@ @@@@@ @@@@@@@@@#
//             @@@@@    %@@@@@    @@@@@@@     @@@@@    .@@@@
//                      %@@@@@                @@@@@
//                      %@@@@@                @@@@@
//
//
//                               who*anon*
//
//
//
//                 @@@@@@&&&&&&&&&&&&&&&&&&&&&&&&&&&&@@@@@@
//          p   m  @@@@@@&&&&&&/      %%&&&%*     (&&&@@@@@
//          o   e  @@@@@@&&&&&&&%.     /%%*       #&&&@@@@@
//          w b t  @@@@@@&&&&&&&%#              ,%&&&@@@@@@
//          e y a  @@@@@@&&&&&&&&&%%,       .    (%&&@@@@@@
//          r   d  @@@@@@&&&&&&&&&&&%%    (/     #&&&@@@@@@
//          e   r  @@@@@@&&&#  &&&&&&&&%%%%*    /%&&&@@@@@@
//          d   o  @@@@@@&&&*   (&&&&&&&&&%.    (&&&@@@@@@@
//              p  @@@@@@&&&&&&&&&&&&&&&&&&&&&&&&&&&&@@@@@@
//
import "@openzeppelin/contracts/token/ERC1155/ERC1155.sol";
import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/security/Pausable.sol";
import "@openzeppelin/contracts/token/ERC1155/extensions/ERC1155Burnable.sol";
import "@openzeppelin/contracts/token/ERC1155/extensions/ERC1155Supply.sol";
import "@chainlink/contracts/src/v0.8/interfaces/AggregatorV3Interface.sol";

/**
 * @dev who*anon* ERC-1155 contract
 */
contract WhoAnonByMetadrop is
  ERC1155,
  Ownable,
  Pausable,
  ERC1155Burnable,
  ERC1155Supply
{
  // ERC-2981: NFT Royalty Standard
  bytes4 private constant _INTERFACE_ID_ERC2981 = 0x2a55205a;

  /**
   * @dev Add name and symbol for consistency with ERC-721 NFTs. Note that ERC-721 stores
   * these variables on-chain, but as they can only be set on the constructor we may as well
   * save the storage and have them as constants in the bytecode. You're welcome Ethereum!
   */
  string private constant NAME = "who*anon*";
  string private constant SYMBOL = "WHOANON";

  /**
   * @dev Developer address:
   */
  address public developer;
  address public beneficiary;

  /**
   * @dev Price buffer above and below the passed amount of ETH that will be accepted. This function
   * will be used to set the price for items in the UI, but there is always the possibility of price
   * fluctuations beween the display and the mint. These parameters determine as an amount per thousand
   * how high above or below the price the passed amount of ETH can be and still make a valid sale. The
   * stored values are in the following format:
   *   - priceBufferUp: amount as a proportion of 1,000. For example, if you set this to 1005 you allow the
   *       price to be up to 1005 / 1000 of the actual price, i.e. not exceeding 0.5% greater.
   *   - priceBufferDown: amount as a proportion of 1,000. For example, if you set this to 995 you allow the
   *       price to be up to 995 / 1000 of the actual price i.e. not exceeding 0.5% less.
   */
  uint16 public priceBufferUp;
  uint16 public priceBufferDown;

  bool public pausableShutoffProtectionDisabled = false;
  bool public pausableDisabled = false;

  bool public publicationShutoffProtectionDisabled = false;
  bool public publicationDisabled = false;

  // while we initialize to 0, we will mint the first token at 1
  uint256 public latestEdition;

  /**
   * @dev ERC-2981 configuration
   */
  address public royaltyReceipientAddress;
  uint256 public royaltyPercentageBasisPoints;

  AggregatorV3Interface internal priceFeed;

  /**
   * @dev titles struct:
   */
  struct PublishedTitle {
    // Slot 1 and 2 (at least)
    string titleURI;
    // Slot 3, 64 + 128 + 64 = 256
    uint64 maxSupply;
    uint128 priceInUSD;
    uint64 startTime;
    // Slot 4, 64 + 64 + 8 + 8 = 144
    uint64 endTime;
    uint64 developerAllocation;
    bool developerAllocationLocked;
    bool exists;
  }

  /**
   * @dev map token classes to parameters:
   */
  mapping(uint256 => PublishedTitle) public publishedTitles;

  mapping(uint256 => uint256) public developerAllocationMinted;

  /**
   * @dev Contract events:
   */
  event PriceBufferUpSet(uint256 priceBuffer);
  event PriceBufferDownSet(uint256 priceBuffer);
  event YouAreAnon(
    address account,
    uint256 tokenId,
    uint256 quantity,
    uint256 cost
  );
  event YouAreRedeemedAnon(
    address account,
    uint256 tokenId,
    uint256 quantity,
    bytes32 hashData,
    bytes data
  );
  event TitlePublished(
    uint256 tokenId,
    string titleURI,
    string redeemableURI_,
    uint64 maxSupply,
    uint128 priceInUSD,
    uint64 mintStartDate,
    uint64 mintEndDate,
    uint64 redeemStartDate,
    uint64 redeemEndDate,
    uint64 developerAllocation,
    bool developerAllocationLocked
  );

  error SupplyExceeded();
  error CannotMintRedeemTokens();
  error CannotRedeemRedeemTokens();
  error InvalidParameters();
  error DeveloperAllocationExceeded();
  error DeveloperAllocationLocked();
  error PublicationIsDisabled();
  error PausableIsDisabled();
  error PublicationShutoffProtectionIsOn();
  error PausableShutoffProtectionIsOn();

  /**
   * @dev Constructor must be passed an array of shareholders for the payment splitter, the first
   * array holding addresses and the second the corresponding shares. For example, you could have the following:
   *   - payees_ [beneficiaryAddress, developerAddress]
   *   - shares_ [90,10]
   * In this example the beneficiary address passed in can claim 90% of total ETH, the developer 10%
   */
  constructor(
    uint16[] memory priceBuffers_,
    address beneficiary_,
    address developer_,
    address priceFeedAddress_,
    PublishedTitle[] memory firstEditions_
  ) ERC1155("") {
    setPriceBufferUp(priceBuffers_[0]);
    setPriceBufferDown(priceBuffers_[1]);
    beneficiary = beneficiary_;
    developer = developer_;
    // @dev Contract address for pricefeed data.
    // MAINNET: 0x5f4eC3Df9cbd43714FE2740f5E3616155c5b8419
    // GOERLI: 0xD4a33860578De61DBAbDc8BFdb98FD742fA7028e
    priceFeed = AggregatorV3Interface(priceFeedAddress_);
    _publishFirstEditions(firstEditions_);
  }

  modifier whenMintingOpen(uint256 tokenId_) {
    require(mintingIsOpen(tokenId_), "Minting is not open");
    _;
  }

  modifier whenMintingClosed(uint256 tokenId_) {
    require(!mintingIsOpen(tokenId_), "Minting is open");
    _;
  }

  modifier onlyOwnerOrDeveloper() {
    require(
      msg.sender == owner() || msg.sender == developer,
      "Only owner or developer"
    );
    _;
  }

  /**
   * @dev whenDeveloperAllocationAvailable
   */
  modifier whenDeveloperAllocationAvailable(
    uint256 tokenId_,
    uint256 quantity_
  ) {
    if (
      (developerAllocationMinted[tokenId_] + quantity_) >
      publishedTitles[tokenId_].developerAllocation
    ) {
      revert DeveloperAllocationExceeded();
    }
    _;
  }

  /**
   *  =======================================
   *  ADMIN FUNCTIONS
   *  =======================================
   */

  /**
   * @dev owner can publish new titles:
   */
  function publish(
    string memory mintableURI_,
    string memory redeemableURI_,
    uint64 maxSupply_,
    uint128 priceInUSD_,
    uint64 mintableStartTime_,
    uint64 mintableDurationInHours_,
    uint64 redeemableStartTime_,
    uint64 redeemableDurationInHours_,
    uint64 developerAllocation_,
    bool developerAllocationLocked_
  ) external onlyOwner {
    if (publicationDisabled) {
      revert PublicationIsDisabled();
    }

    uint256 mintableEditionTokenId = latestEdition + 1;
    uint256 redeemableEditionTokenId = mintableEditionTokenId + 1;

    // Publish the mintable edition:
    publishedTitles[mintableEditionTokenId].titleURI = mintableURI_;
    publishedTitles[mintableEditionTokenId].maxSupply = maxSupply_;
    publishedTitles[mintableEditionTokenId].priceInUSD = priceInUSD_;
    publishedTitles[mintableEditionTokenId].startTime = mintableStartTime_;
    publishedTitles[mintableEditionTokenId].endTime =
      mintableStartTime_ +
      (mintableDurationInHours_ * 1 hours);
    publishedTitles[mintableEditionTokenId]
      .developerAllocation = developerAllocation_;
    publishedTitles[mintableEditionTokenId]
      .developerAllocationLocked = developerAllocationLocked_;
    publishedTitles[mintableEditionTokenId].exists = true;

    // Publish the redeemable edition:
    publishedTitles[redeemableEditionTokenId].titleURI = redeemableURI_;
    publishedTitles[redeemableEditionTokenId].maxSupply = 0;
    publishedTitles[redeemableEditionTokenId].priceInUSD = 0;
    publishedTitles[redeemableEditionTokenId].startTime = redeemableStartTime_;
    publishedTitles[redeemableEditionTokenId].endTime =
      redeemableStartTime_ +
      (redeemableDurationInHours_ * 1 hours);
    publishedTitles[redeemableEditionTokenId].developerAllocation = 0;
    publishedTitles[redeemableEditionTokenId].developerAllocationLocked = true;
    publishedTitles[redeemableEditionTokenId].exists = true;

    latestEdition = redeemableEditionTokenId;

    emit TitlePublished(
      mintableEditionTokenId,
      mintableURI_,
      redeemableURI_,
      maxSupply_,
      priceInUSD_,
      mintableStartTime_,
      mintableStartTime_ + (mintableDurationInHours_ * 1 hours),
      redeemableStartTime_,
      redeemableStartTime_ + (redeemableDurationInHours_ * 1 hours),
      developerAllocation_,
      developerAllocationLocked_
    );
  }

  /**
   * @dev updatePriceFeedAddress
   */
  function updatePriceFeedAddress(address priceFeedAddress_)
    external
    onlyOwner
  {
    priceFeed = AggregatorV3Interface(priceFeedAddress_);
  }

  /**
   * @dev updateLatestEdition
   */
  function updateLatestEdition(uint32 latestEdition_) external onlyOwner {
    latestEdition = latestEdition_;
  }

  /**
   * @dev disablePausableShutoffProtection
   */
  function disablePausableShutoffProtection() external onlyOwner {
    pausableShutoffProtectionDisabled = true;
  }

  /**
   * @dev disablePausable
   */
  function disablePausable() external onlyOwner {
    if (pausableShutoffProtectionDisabled) {
      pausableDisabled = true;
    } else {
      revert PausableShutoffProtectionIsOn();
    }
  }

  /**
   * @dev enablePausableShutoffProtection
   */
  function enablePausableShutoffProtection() external onlyOwner {
    pausableShutoffProtectionDisabled = false;
  }

  /**
   * @dev disablePausableShutoffProtection
   */
  function disablePublicationShutoffProtection() external onlyOwner {
    publicationShutoffProtectionDisabled = true;
  }

  /**
   * @dev disablePublication
   */
  function disablePublication() external onlyOwner {
    if (publicationShutoffProtectionDisabled) {
      publicationDisabled = true;
    } else {
      revert PublicationShutoffProtectionIsOn();
    }
  }

  /**
   * @dev enablePublicationShutoffProtection
   */
  function enablePublicationShutoffProtection() external onlyOwner {
    publicationShutoffProtectionDisabled = false;
  }

  /**
   * @dev owner can reduce supply:
   */
  function reduceSupply(uint256 tokenId_, uint64 maxSupply_)
    external
    onlyOwner
  {
    require(publishedTitles[tokenId_].exists, "Token ID does not exist");

    // A supply of 0 is unlimited, so under no circumstances can this be a valid update:
    require(
      maxSupply_ != 0,
      "Cannot set to unlimited after initial publication"
    );

    // A supply of 0 is unlimited, so always allow a reduction from unlimited:
    if (publishedTitles[tokenId_].maxSupply > 0) {
      require(
        publishedTitles[tokenId_].maxSupply > maxSupply_,
        "Supply can only be decreased"
      );
    }
    publishedTitles[tokenId_].maxSupply = maxSupply_;
  }

  /**
   * @dev updateTokenURI
   */
  function updateTokenURI(uint256 tokenId_, string memory uri_)
    external
    onlyOwner
  {
    require(publishedTitles[tokenId_].exists, "Token ID does not exist");

    publishedTitles[tokenId_].titleURI = uri_;
  }

  /**
   * @dev updateTokenPriceInUSD
   */
  function updateTokenPriceInUSD(uint256 tokenId_, uint128 tokenPrice_)
    external
    onlyOwner
  {
    require(publishedTitles[tokenId_].exists, "Token ID does not exist");

    // A supply of 0 is unlimited, so under no circumstances can this be a valid update:
    publishedTitles[tokenId_].priceInUSD = tokenPrice_;
  }

  /**
   * @dev updateStartTime
   */
  function updateStartTime(uint256 tokenId_, uint64 startTime_)
    external
    onlyOwner
  {
    require(publishedTitles[tokenId_].exists, "Token ID does not exist");

    publishedTitles[tokenId_].startTime = startTime_;
  }

  /**
   * @dev updateEndTime
   */
  function updateEndTime(uint256 tokenId_, uint64 endTime_) external onlyOwner {
    require(publishedTitles[tokenId_].exists, "Token ID does not exist");

    publishedTitles[tokenId_].endTime = endTime_;
  }

  /**
   * @dev updateDeveloperAllocation
   */
  function updateDeveloperAllocation(
    uint256 tokenId_,
    uint64 developerAlloaction_
  ) external onlyOwner {
    require(publishedTitles[tokenId_].exists, "Token ID does not exist");

    if (publishedTitles[tokenId_].developerAllocationLocked) {
      revert DeveloperAllocationLocked();
    }

    publishedTitles[tokenId_].developerAllocation = developerAlloaction_;
  }

  /**
   * @dev lockDeveloperAllocationForTokenId
   */
  function lockDeveloperAllocationForTokenId(uint256 tokenId_)
    external
    onlyOwner
  {
    publishedTitles[tokenId_].developerAllocationLocked = true;
  }

  /**
   * @dev updateDeveloper
   */
  function updateDeveloper(address developer_) external onlyOwner {
    developer = developer_;
  }

  /**
   * @dev updateBeneficiary
   */
  function updateBeneficiary(address beneficiary_) external onlyOwner {
    beneficiary = beneficiary_;
  }

  /**
   * @dev setPriceBufferUp
   */
  function setPriceBufferUp(uint16 priceBufferUpToSet_) public onlyOwner {
    priceBufferUp = priceBufferUpToSet_;
    emit PriceBufferUpSet(priceBufferUp);
  }

  /**
   * @dev setPriceBufferDown
   */
  function setPriceBufferDown(uint16 priceBufferDownToSet_) public onlyOwner {
    priceBufferDown = priceBufferDownToSet_;
    emit PriceBufferDownSet(priceBufferDown);
  }

  /**
   * @dev pause
   */
  function pause() public onlyOwner {
    if (pausableDisabled) {
      revert PausableIsDisabled();
    }
    _pause();
  }

  /**
   * @dev unpause
   */
  function unpause() public onlyOwner {
    _unpause();
  }

  /**
   *  =======================================
   *  GETTERS
   *  =======================================
   */

  /**
   * @dev get Title details
   */
  function getTitleDetails(uint256 _tokenId)
    external
    view
    returns (PublishedTitle memory)
  {
    return (publishedTitles[_tokenId]);
  }

  function getPriceFeedAddress() external view returns (address) {
    return (address(priceFeed));
  }

  function mintingIsOpen(uint256 tokenId_) public view returns (bool) {
    return (block.timestamp >= publishedTitles[tokenId_].startTime &&
      block.timestamp <= publishedTitles[tokenId_].endTime);
  }

  function getCurrentRate() external view returns (uint256) {
    return (uint256(getLatestPrice()));
  }

  function getDollarValueInWei(uint256 _dollarValue)
    external
    view
    returns (uint256)
  {
    uint256 latestPrice = uint256(getLatestPrice());
    return (_performConversion(latestPrice, _dollarValue));
  }

  function getETHPriceForTokenId(uint256 tokenId_)
    public
    view
    returns (uint256 price_)
  {
    uint256 latestPrice = uint256(getLatestPrice());

    return (
      _performConversion(latestPrice, publishedTitles[tokenId_].priceInUSD)
    );
  }

  function getBuffers()
    external
    view
    onlyOwner
    returns (uint256 bufferUp_, uint256 bufferDown_)
  {
    return (priceBufferUp, priceBufferDown);
  }

  /**
   * @dev Add name, symbol and total supply for consistency with ERC-721 NFTs.
   */
  function name() public pure returns (string memory) {
    return NAME;
  }

  function symbol() public pure returns (string memory) {
    return SYMBOL;
  }

  function totalSupply()
    public
    view
    returns (uint256 totalSupplyForAllCollections_)
  {
    for (uint256 i = 1; i <= latestEdition; ) {
      totalSupplyForAllCollections_ += totalSupply(i);
      unchecked {
        i++;
      }
    }
    return totalSupplyForAllCollections_;
  }

  /**
   * Returns the latest USD price to 8DP of 1 ETH
   */
  function getLatestPrice() public view returns (int256) {
    (, int256 price, , , ) = priceFeed.latestRoundData();
    return price;
  }

  /**
   * @dev _publishFirstEditions
   */
  function _publishFirstEditions(PublishedTitle[] memory firstEditions_)
    internal
  {
    for (uint256 i = 0; i < firstEditions_.length; ) {
      publishedTitles[i + 1] = firstEditions_[i];

      unchecked {
        i++;
      }
    }
    for (uint256 i = 0; i < firstEditions_.length; ) {
      emit TitlePublished(
        i + 1, // unredeemed edition
        firstEditions_[i].titleURI,
        firstEditions_[i + 1].titleURI,
        firstEditions_[i].maxSupply,
        firstEditions_[i].priceInUSD,
        firstEditions_[i].startTime,
        firstEditions_[i].endTime,
        firstEditions_[i + 1].startTime,
        firstEditions_[i + 1].endTime,
        firstEditions_[i].developerAllocation,
        firstEditions_[i].developerAllocationLocked
      );
      unchecked {
        i += 2;
      }
    }
    // Update the latest Edition number:
    latestEdition = firstEditions_.length;
  }

  /**
   * @dev perform price conversion USD to Wei at the prescribed number of significant figures (i.e. DP in ETH)
   */
  function _performConversion(uint256 _price, uint256 _value)
    internal
    pure
    returns (uint256 convertedValue)
  {
    require(_price > 0 && _price < 9999999999999, "Pricing Error");
    // The USD figure from the price feed is one eth in USD to 8 DP. We need the value of one dollar in wei/
    // The price feed has 8DP so lets add that exponent to our wei figure to give us the value of $1 in wei
    uint256 oneUSDInWei = ((10**26) / _price);
    // 2) Mutiply our dollar value by that to get our value in wei:
    uint256 valueInWei = oneUSDInWei * _value;

    // 3) And then roundup that number to 4DP of eth by removing 10**14 digits, adding 1, then multiplying by 10**14:
    valueInWei = ((valueInWei / (10**14)) + 1) * (10**14);
    return (valueInWei);
  }

  /**
   * @dev This function is called from the UI to mint NFTs for the user.
   */
  function iAmAnon(uint256 tokenId_, uint256 quantity_)
    external
    payable
    whenMintingOpen(tokenId_)
    whenNotPaused
  {
    require(quantity_ != 0, "Order must be for an item");

    // Can only mint mintables, i.e. ODD numbered token IDs. The even equivalent (minted 1 = redeemed 2 etc)
    // must be obtained on redemption of the minted
    if (!_isMintableToken(tokenId_)) {
      revert CannotMintRedeemTokens();
    }

    // Check that we aren't requesting more than is available:
    if (_orderExceedsSupply(tokenId_, quantity_)) {
      revert SupplyExceeded();
    }

    // Calculate the required price for this order:
    uint256 orderPrice = _priceOrder(tokenId_, quantity_);

    // Check the payment is correct:
    _checkPaymentToPrice(msg.value, orderPrice);

    // To reach here the price and quantity check must have passed. Mint the items:
    _mint(msg.sender, tokenId_, quantity_, "");

    emit YouAreAnon(msg.sender, tokenId_, quantity_, msg.value);
  }

  /**
   * @dev This function is called on redemption
   */
  function redeemAnon(
    uint256 tokenId_,
    uint256 quantity_,
    bytes32 dataHash_,
    bytes memory data_
  ) external whenMintingOpen(tokenId_ + 1) whenNotPaused {
    if (_isRedemptionToken(tokenId_)) {
      revert CannotRedeemRedeemTokens();
    }

    require(quantity_ != 0, "Redemption must be for an item");

    burn(msg.sender, tokenId_, quantity_);

    // All good so far? OK, mint their redeption equivalents
    _mint(msg.sender, (tokenId_ + 1), quantity_, "");

    emit YouAreRedeemedAnon(msg.sender, tokenId_, quantity_, dataHash_, data_);
  }

  /**
   * @dev _orderExceedsSupply
   */
  function _orderExceedsSupply(uint256 tokenId_, uint256 quantity_)
    internal
    view
    returns (bool)
  {
    // Note a supply set to 0 is unlimited:
    uint256 maxSupply = publishedTitles[tokenId_].maxSupply;

    if ((maxSupply != 0) && ((quantity_ + totalSupply(tokenId_)) > maxSupply)) {
      return true;
    } else {
      return false;
    }
  }

  /**
   * @dev _isMintableToken
   */
  function _isMintableToken(uint256 tokenId_) internal pure returns (bool) {
    return (tokenId_ % 2 != 0);
  }

  /**
   * @dev _isRedepmtionToken
   */
  function _isRedemptionToken(uint256 tokenId_) internal pure returns (bool) {
    return (tokenId_ % 2 == 0);
  }

  /**
   * @dev Get the current price of this order in the same way that it will have been assembled in the UI,
   * i.e. get the current price of each token type in ETH (including the rounding to 4DP of ETH) and then
   * multiply that by the total quantity ordered.
   */
  function _priceOrder(uint256 tokenId_, uint256 quantity_)
    internal
    view
    returns (uint256 price)
  {
    uint256 orderCostInETH = 0;

    uint256 unitPrice = getETHPriceForTokenId(tokenId_);

    orderCostInETH = (unitPrice * quantity_);

    return (orderCostInETH);
  }

  /**
   * @dev mintDeveloperAllocation
   */
  function mintDeveloperAllocation(uint256 tokenId_, uint256 quantity_)
    external
    payable
    onlyOwnerOrDeveloper
    whenDeveloperAllocationAvailable(tokenId_, quantity_)
  {
    _mint(developer, tokenId_, quantity_, "");

    developerAllocationMinted[tokenId_] += quantity_;
  }

  /**
   * @dev Determine if the passed cost is within bounds of current price:
   */
  function _checkPaymentToPrice(uint256 _passedETH, uint256 _orderPrice)
    internal
    view
  {
    // Establish upper and lower bands of price buffer and check
    uint256 orderPriceLower = (_orderPrice * priceBufferDown) / 1000;

    require(_passedETH >= orderPriceLower, "Insufficient ETH passed for order");

    uint256 orderPriceUpper = (_orderPrice * priceBufferUp) / 1000;

    require(_passedETH <= orderPriceUpper, "Too much ETH passed for order");
  }

  /**
   * @dev
   */
  function setRoyaltyPercentageBasisPoints(
    uint256 royaltyPercentageBasisPoints_
  ) external onlyOwner {
    royaltyPercentageBasisPoints = royaltyPercentageBasisPoints_;
  }

  /**
   * @dev
   */
  function setRoyaltyReceipientAddress(
    address payable royaltyReceipientAddress_
  ) external onlyOwner {
    royaltyReceipientAddress = royaltyReceipientAddress_;
  }

  /**
   * @dev
   */
  function royaltyInfo(uint256, uint256 salePrice_)
    external
    view
    returns (address receiver, uint256 royaltyAmount)
  {
    uint256 royalty = (salePrice_ * royaltyPercentageBasisPoints) / 10000;
    return (royaltyReceipientAddress, royalty);
  }

  /**
   * @dev
   */
  function supportsInterface(bytes4 interfaceId)
    public
    view
    override(ERC1155)
    returns (bool)
  {
    return
      interfaceId == _INTERFACE_ID_ERC2981 ||
      super.supportsInterface(interfaceId);
  }

  /**
   *  =======================================
   *  STANDARD FUNCTIONS
   *  =======================================
   */
  function uri(uint256 _tokenId)
    public
    view
    virtual
    override
    returns (string memory)
  {
    return publishedTitles[_tokenId].titleURI;
  }

  /**
   *
   * @dev withdrawContractBalance: onlyOwner withdrawal to the beneficiary address
   *
   */
  function withdrawContractBalance() external onlyOwner {
    (bool success, ) = beneficiary.call{value: address(this).balance}("");
    require(success, "Transfer failed");
  }

  /**
   *
   * @dev withdrawETH: onlyOwner withdrawal to the beneficiary address, sending
   * the amount to withdraw as an argument
   *
   */
  function withdrawETH(uint256 amount_) external onlyOwner {
    (bool success, ) = beneficiary.call{value: amount_}("");
    require(success, "Transfer failed");
  }

  /**
   * @dev The fallback function is executed on a call to the contract if
   * none of the other functions match the given function signature.
   */
  fallback() external payable {
    revert();
  }

  /**
   * @dev revert any random ETH:
   */
  receive() external payable {
    revert();
  }

  function _beforeTokenTransfer(
    address operator,
    address from,
    address to,
    uint256[] memory ids,
    uint256[] memory amounts,
    bytes memory data
  ) internal override(ERC1155, ERC1155Supply) whenNotPaused {
    super._beforeTokenTransfer(operator, from, to, ids, amounts, data);
  }
}